

```

1 // Key.as
2 package {
3
4     import flash.display.Stage;
5     import flash.events.Event;
6     import flash.events.KeyboardEvent;
7
8     /**
9      * The Key class recreates functionality of
10     * Key.isDown of ActionScript 1 and 2. Before using
11     * Key.isDown, you first need to initialize the
12     * Key class with a reference to the stage using
13     * its Key.initialize() method. For key
14     * codes use the flash.ui.Keyboard class.
15     *
16     * Usage:
17     * Key.initialize(stage);
18     * if (Key.isDown(Keyboard.LEFT)) {
19     *     // Left key is being pressed
20     * }
21     */
22     public class Key {
23
24         private static var initialized:Boolean = false; // marks whether or not the class has been initialized
25         private static var keysDown:Object = new Object(); // stores key codes of all keys pressed
26
27         /**
28          * Initializes the key class creating assigning event
29          * handlers to capture necessary key events from the stage
30          */
31         public static function initialize(stage:Stage) {
32             if (!initialized) {
33                 // assign listeners for key presses and deactivation of the player
34                 stage.addEventListener(KeyboardEvent.KEY_DOWN, keyPressed);
35                 stage.addEventListener(KeyboardEvent.KEY_UP, keyReleased);
36                 stage.addEventListener(Event.DEACTIVATE, clearKeys);
37
38                 // mark initialization as true so redundant
39                 // calls do not reassign the event handlers
40                 initialized = true;
41             }
42         }
43
44         /**
45          * Returns true or false if the key represented by the
46          * keyCode passed is being pressed
47          */
48         public static function isDown(keyCode:uint):Boolean {
49             if (!initialized) {
50                 // throw an error if isDown is used
51                 // prior to Key class initialization
52                 throw new Error("Key class has yet been initialized.");
53             }
54             return Boolean(keyCode in keysDown);
55         }
56
57         /**
58          * Event handler for capturing keys being pressed
59          */
60         private static function keyPressed(event:KeyboardEvent):void {
61             // create a property in keysDown with the name of the keyCode
62             keysDown[event.keyCode] = true;
63         }
64
65         /**
66          * Event handler for capturing keys being released
67          */
68         private static function keyReleased(event:KeyboardEvent):void {
69             if (event.keyCode in keysDown) {
70                 // delete the property in keysDown if it exists
71                 delete keysDown[event.keyCode];
72             }
73         }
74
75         /**
76          * Event handler for Flash Player deactivation
77          */
78         private static function clearKeys(event:Event):void {
79             // clear all keys in keysDown since the player cannot
80             // detect keys being pressed or released when not focused
81             keysDown = new Object();
82         }
83     }
84 }

```